===PhaseLock_Loop==========



```
.OPTIONS  GMIN=1e-12    METHOD=trap   srcsteps = 1  gminsteps = 1
*MUX===== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VCC    0              DC     4.9
VDC1      VDC1   0              DC     2.2
IDC1      VEMIT1 0              DC     1m
VAC1      VAC1   VDC1           SIN(   0      20m    200         )
QN1       VEMIT2 FBP    VEMIT1         NPN1   1
QN2       VEMIT3 0      VEMIT1         NPN1   1
QN3       OUT1   VAC1   VEMIT2         NPN1   1
QN4       OUT2   VDC1   VEMIT2         NPN1   1
QN5       OUT1   VDC1   VEMIT3         NPN1   1
QN6       OUT2   VAC1   VEMIT3         NPN1   1
*FILTER== ====== ====== ====== ====== ====== ====== ====== ======
R1F       VCC    OUT1   1k
R2F       VCC    OUT2   1k
C2F       OUT1   OUT2   2u
*VCO===== ====== ====== ====== ====== ====== ====== ====== ======
E_GAIN1   VFC    VDC2   OUT1   OUT2   5
VDC2      VDC2   0              DC     1.2
VEE       VEE    0              DC    -4.9
V_in      VIN    0              PULSE( 0  100m  1m 1u 1u 1m 1 )
G_GM1     VGAIN  0      FBP    FBN    10m
D_1       VGAIN  VCC                   DMODEL
D_2       VEE    VGAIN                 DMODEL
MP1       OUT    VGAIN  VCC    VCC    PMOSL1 L=1u   W=30u
MN1       OUT    VGAIN  VEE    VEE    NMOSL1 L=1u   W=30u
MN2       FBP    VFC    VIN    VIN    NMOSL1 L=1u   W=1u
R1        VIN    FBP    10K
R2        FBP    OUT    100k
C1        FBN    0      2u
R3        FBN    OUT    10k
```

```
.control
op
tran     100u   100m    0
run
plot     50*(vac1-vdc1) fbp fbn out vfc
.endc

*======== ====== ====== ====== ====== ====== ====== ====== ======
.model    NPN1   NPN
.model    SMODEL SW     ron=1000
.model    DMODEL D
**** LEVEL 1     NMOS ****
.MODEL            NMOSL1          NMOS
+ LEVEL=1         TPG=1           TOX=1.9800000E-08  NSUB=
4.9999999E+16
+ VT0=0.69486     GAMMA=0.60309   KP=2.33082E-05     LAMBDA=0.013333
+ PHI=1           LD=0.1U         NSS=0.0000000E+00
+ CJ=4.091E-4     MJ=0.307        PB=1.0
+ CJSW=3.078E-10 MJSW=1.0E-2
+ CGSO=3.93E-10   CGDO=3.93E-10
**** LEVEL 1     PMOS ****
.MODEL            PMOSL1          PMOS
+ LEVEL=1         TPG=-1          TOX=1.9800000E-08  NSUB=
4.9999999E+17
+ VT0=-0.60865    GAMMA=0.89213   KP=7.69968E-06     LAMBDA=0.018966
+ PHI=1           LD=0.28U        NSS=0.0000000E+00
+ J=6.852E-4     MJ=0.429        PB=1.0
+ CJSW=5.217E-10 MJSW=0.351
+ CGSO=7.29E-10   CGDO=7.29E-10

.end
```


Graph 1 - tran2: ===PhaseLock_Loop==========
V
50*(vac1-vdc1)
fbn
vfc
fbp
out
6.0
4.0
2.0
0.0
-2.0
-4.0
-6.0
voltage
0.0
20.0
40.0
60.0
80.0
time ms